INTERNATIONAL JOURNAL OF QUANTUM CHEMISTRY, VOL. 52, 593-607 (1994)

Computation of Algebraic Formulas for Wigner 3-j, 6-j, and 9-j Symbols by Maple

SHAN-TAO LAI

Vitreous State Laboratory and Department of Chemistry, The Catholic University of America, Washington, DC 20064, and Department of Chemistry, Xiamen University, Xiamen, Fujian 361005, People's Republic of China

Abstract

Three programs capable of computing algebraic formulas for Wigner 3-j, 6-j, and 9-j symbols have been written in the Maple* program language. These programs can also compute numerically exact values of 3-j, 6-j, and 9-j symbols.

1. Introduction

In a previous article, we reported numerical exact computations of the Wigner 3-j, 6-j, and 9-j symbols by a symbolic mathematical software system: Derive, Maple, and Reduce [1]. In this article, we use the Maple* V software system to compute the algebraic formulas of 3-j, 6-j, and 9-j symbols. The results agree with the books of Zare [2], Varshalovich et al. [3], Edmonds [4], and Biedenharn and Louck [5]. The programs apply to both integer and half-integer angular momentum quantum numbers. They can compute numerically exact and floating-point values of 3-j, 6-j, and 9-j symbols. The proposed programs can be easily translated into Reduce,† Mathematica,‡ and Macsyma,§ and other program languages. Mathematica version 2.2 has included the computation of symbolic parameters in 3-j and 6-j symbols.

2. Methods

A. 3-j Symbol

In the literature, there are four different ways to obtain two angular momenta coupling expressions [6–9]. Here, we modify Wigner's formula as

$$\begin{pmatrix} j_1 & j & j_2 \\ m_1 & m & m_2 \end{pmatrix} = \Delta(j_1, j_2, j)\sqrt{\frac{(j_2 - m_2)!\,(j_2 + m_2)!}{(j + m)!\,(j - m)!\,(j_1 - m - m_2)!\,(j_1 + m + m_2)!}}$$
$$\times \sum_z (-1)^{2j - j_1 - m_1 + z} \frac{(j + j_2 - m - m_2 - z)!\,(j_1 + m + m_2 + z)!}{z!\,(j_2 - m_2 - z)!\,(u - z)!\,(j_2 + m_2 - u + z)!},$$

*Maple is a registered trademark of Waterloo Maple Software.

†Reduce is a registered trademark of The Rand Corporation, Santa Monica, CA 90406.

‡Mathematica is a registered trademark of Wolfram Research Inc., Champaign, IL 61820.

§Macsyma is a registered trademark of Macsyma Inc., Arlington, MA 02174.

 CCC 0020-7608/94/030593-15

(1)

where

$$u = j - j_1 + j_2, \tag{2}$$

and

$$\Delta(j_1, j_2, j) = \sqrt{\frac{(j_1 + j - j_2)!(j_1 - j + j_2)!(-j_1 + j + j_2)!}{(j_1 + j + j_2 + 1)!}}. \tag{3}$$

In Eq. (1), the summation over z extends over all values of the variable satisfying the inequality

$$0 \leq z \leq min(j_2 - m_2, u). \tag{4}$$

Clearly, if the value of j_2 is given, j can assume the values from $|j_1 - j_2|$ to $j_1 + j_2$ for all possible values of m_2 from $-j_2$ to j_2, i.e.,

$$|j_1 - j_2| \leq j \leq j_1 + j_2 \tag{5}$$

and

$$-j_2 \leq m_2 \leq j_2. \tag{6}$$

A Maple V program $3jfunm$ has been written based on the above Eqs. (1)–(6).

*B. 6-*j *Symbol*

We use Jucys and Bandzaitis' 6-j formula [10] and use the properties of 6-j and obtain

$$\begin{Bmatrix} e & a & f \\ b & d & c \end{Bmatrix} = \frac{(-1)^{b+c+e+f}\Delta(abc)\Delta(aef)\Delta(cde)\Delta(bdf)(a + b + c + 1)!(b + d + f + 1)!}{(a + b - c)!(c - d + e)!(c + d - e)!(a - e + f)!(-a + e + f)!(b + d - f)!}$$
$$\times \sum_z \frac{(-1)^z(2b - z)!(b + c - e + f - z)!(b + c + e + f + 1 - z)!}{z!(-a + b + c - z)!(b - d + f - z)!(a + b + c + 1 - z)!(b + d + f + 1 - z)!}, \tag{7}$$

where $\Delta(abc)$ is defined in Eq. (3). In Eq. (7), the summation over z must satisfy the inequality

$$0 \leq z \leq min(2b, -a + b + c, b - d + f). \tag{8}$$

If the value of $b \geq 0$ is given, and $d = f + \mu$ and $c = a + \nu$, where $|\mu|, |\nu| \leq b$ and b, μ, ν can be integers or half-integers, then the algebraic formulas of 6-j symbols can be computed from Eq. (7). A Maple V program $6jfunm$ has been written based on the above Eqs. (7) and (8).

C. 9-j *Symbol*

We use the 9-j formula of Jucys and Bandzaitis [10] and the properties of 9-j symbols to obtain

$$\begin{pmatrix} j_{11} & j_{21} & j_{31} \\ j_{12} & j_{22} & j_{32} \\ j_{13} & j_{23} & j_{33} \end{pmatrix} = (-1)^{j_{13}+j_{23}-j_{33}} \frac{\nabla(j_{21}j_{11}j_{31})\nabla(j_{12}j_{22}j_{32})\nabla(j_{33}j_{31}j_{32})}{\nabla(j_{21}j_{22}j_{23})\nabla(j_{12}j_{11}j_{13})\nabla(j_{33}j_{13}j_{23})}$$
$$\times \sum_{x,y,z} \frac{(-1)^{x+y+z}(2j_{23}-x)!\,(j_{21}+j_{22}-j_{23}+x)!}{x!\,(j_{22}-j_{21}+j_{23}-x)!\,(j_{13}+j_{23}-j_{33}-x)!\,(j_{21}-j_{12}+j_{32}-j_{23}+x+y)!}$$
$$\times \frac{(j_{13}-j_{23}+j_{33}+x)!\,(j_{22}-j_{12}+j_{32}+y)!}{(j_{12}-j_{11}-j_{23}+j_{33}+x+z)!\,y!\,(j_{12}+j_{22}-j_{32}-y)!}$$
$$\times \frac{(j_{31}+j_{32}-j_{33}+y)!\,(j_{11}+j_{21}-j_{32}+j_{33}-y-z)!}{(j_{31}-j_{32}+j_{33}-y)!\,(2j_{32}+1+y)!\,z!\,(j_{11}+j_{21}-j_{31}-z)!}$$
$$\times \frac{(2j_{11}-z)!\,(j_{12}-j_{11}+j_{13}+z)!}{(j_{11}-j_{12}+j_{13}-z)!\,(j_{11}+j_{21}+j_{31}+1-z)!}, \tag{9}$$

where

$$\nabla(abc) = \sqrt{\frac{(a-b+c)!\,(a+b-c)!\,(a+b+c+1)!}{(b+c-a)!}}. \tag{10}$$

In Eq. (9), the summations over x, y, and z extend over all values satisfying the inequalities

$$0 \le x \le min(2j_{33}, j_{22}-j_{21}+j_{23}, j_{13}+j_{23}-j_{33}), \tag{11}$$
$$0 \le y \le j_{31}-j_{32}+j_{33}, \tag{12}$$
$$max(j_{11}-j_{12}-j_{13}, 0) \le z \le j_{11}-j_{12}+j_{13}, \tag{13}$$
$$0 \le -j_{11}+j_{12}+j_{33}-j_{23}+x+z, \tag{14}$$

From Eqs. (11)–(14), if the values of j_{13}, j_{23}, and j_{33} are given, and $j_{11} = j_{12} + \lambda$, $j_{21} = j_{22} + \mu$ and $j_{31} = j_{32} + \nu$, the algebraic expressions of the 9-j symbol (we use a, b, and c for j_{12}, j_{22}, and j_{32} and α, β, and γ for j_{13}, j_{23}, and j_{33}, respectively):

$$\begin{pmatrix} a+\lambda & b+\mu & c+\nu \\ a & b & c \\ \alpha & \beta & \gamma \end{pmatrix}$$

can be computed by the Maple system. Here, $0 \le \alpha$, $0 \le \beta$, $0 \le \gamma$, and $-\alpha \le \lambda \le \alpha$, $-\beta \le \mu \le \beta$, and $-\gamma \le \nu \le \gamma$, where α, β, and γ can be integers or half-integers. A Maple program $9jfunm$ has been written based on Eqs. (9)–(14).

3. The Programs

A. Input and Output for 3-j Symbols

In this 3-j program, j_2 and m_2 must be both integers or half-integers. For example, we compute

$$\begin{pmatrix} j+\frac{3}{2} & j & \frac{3}{2} \\ m & -m-\frac{1}{2} & \frac{1}{2} \end{pmatrix},$$

$$\begin{pmatrix} j+1 & j & 2 \\ m & -m-1 & 1 \end{pmatrix},$$

and

$$\begin{pmatrix} 16 & 16 & 16 \\ 0 & 0 & 0 \end{pmatrix}.$$

In the Maple V system (all inputs are case sensitive):

```
> read'3jfunm';
w3f:= proc(j1,j,j2,m1,m,m2)
local delta,a,s0,u,v,uv,s;
  if m1+m2+m = 0 then
    delta := sqrt((j1+j2-j)!*(j1-j2+j)!*(-j1+j2+j)!/
      (j1+j2+j+1)!);
    a := sqrt((j2+m2)!*(j2-m2)!/(j+m)!/(j-m)!/(j1-m-m2)!/
      (j1+m+m2)!);
    s0 := 0;
    u := -j1+j2+j;
    v := j2-m2;
    uv := min(u,v);
    if type(j1,numeric) and type(j2,numeric)and
      type(j,numeric) and type(m1,numeric) and
      type(m2,numeric) and type(m,numeric) then
      z1 := max(-j1-m-m2,u-j2-m2,0);
      z2 := min(j+j2-m-m2,j2-m2,u);
      for z from z1 to z2 do
        if 0 <= j+j2-m-m2-z and 0 <= j2-m2-z and
          0 <= u-z and 0 <= j2+m2-u+z
          and 0 <= j1+m+m2+z then
          s0 := s0+(-1)^(2*j-j1-m1+z)*(j+j2-m-m2-z)!*
            (j1+m+m2+z)!/z!/(j2-m2-z)!/(u-z)!/(j2+m2-u+z)!
        fi
      od;
      s := a*delta*s0;
      s := simplify(s)
    else
      for z from 0 to uv do
```

```
        if j2+m2-u+z >=0 then
        s0 := s0+(-1)^(2*j-j1-m1+z)*(j+j2-m-m2-z)!*
          (j1+m+m2+z)!/z!/(j2-m2-z)!/(u-z)!/(j2+m2-u+z)!
        fi

      od;
      s := a*delta*s0;
      s := factor(expand(s))
    fi
  else s := 0
  fi
end
> w3f(j+3/2,j,3/2,m,-m-1/2,1/2);

       1/2                1/2                1/2                 1/2 1/2   j
     I2   (4 j - 4 m + 2)    (4 j + 6 + 4 m)    (4 j + 6 - 4 m)    6   (-1)
1/16----------------------------------------------------------------------
                   1/2            1/2            1/2            1/2      m
        (2 j + 1)     (2 j + 2)      (2 j + 3)      (2 j + 4)      (-1)

> w3f(j+1,j,2,m,-m-1,1);
                                        1/2        1/2  1/2       j
              (j + 2 m + 2) (j + 1 - m)    (j - m)     2     (-1)
   - ------------------------------------------------------------
          m          1/2           1/2          1/2          1/2  1/2
      (-1)  (2 j + 4)    (2 j + 3)     (2 j + 2)    (2 j + 1)     j

> w3f(16,16,16,0,0,0);
   1287        1/2  1/2   1/2   1/2   1/2   1/2   1/2   1/2   1/2
-----------  2     5    11    29    31    37    41    43    47
19293438101

> ifactor(op(1,"));
                              2
                           (3)  (11)(13)
                    ---------------------------
                    (7)(29)(31)(37)(41)(43)(47)
```

The results agree with those reported in the literature [1–3,5]. Additional examples are given in the Appendix. Notice that when j_2 is sufficiently large, we can delete the *expand* command in the program for the 3-j symbol and avoid expansion of the factorial. For example,

$$\begin{pmatrix} j+9 & j & 9 \\ m & -m-1 & 1 \end{pmatrix}$$

with the *expand* command yields the output

```
> w3f(j+9,j,9,m,-m-1,1);
           1/2         1/2            1/2            1/2             1/2
  - 3/5 10    (j - m)    (j + m + 2)    (j + m + 3)    (j + m + 4)
```

```
              1/2              1/2              1/2              1/2              1/2
(j + m + 5)      (j + m + 6)      (j + m + 7)      (j + m + 8)      (j + 9 + m)

              1/2              1/2              1/2              1/2              1/2
(j - m + 1)      (j - m + 2)      (j - m + 3)      (j - m + 4)      (j - m + 5)

             1/2             1/2             1/2             1/2      1/2      j
(j - m + 6)     (j - m + 7)     (j - m + 8)     (j + 9 - m)     12155     (-1)

    /             1/2            1/2            1/2            1/2            1/2
   /   ((2 j + 1)     (2 j + 2)      (2 j + 3)      (2 j + 4)      (2 j + 5)
  /

              1/2            1/2            1/2            1/2             1/2
     (2 j + 6)      (2 j + 7)      (2 j + 8)      (2 j + 9)      (2 j + 10)

                1/2             1/2             1/2             1/2             1/2
   (2 j + 11)      (2 j + 12)      (2 j + 13)      (2 j + 14)      (2 j + 15)

                  1/2             1/2             1/2             1/2     m
      (2 j + 16)      (2 j + 17)      (2 j + 18)      (2 j + 19)      (-1))
```

On the other hand, without the *expand* command, the output is

```
> w3f(j+9,j,9,m,-m-1,1);
3/5

 1/2               1/2               1/2       1/2       1/2    (j - 9 - m)
10   ((j + 9 + m)!)    ((j + 9 - m)!)    12155     ((2 j)!)    (-1)
-----------------------------------------------------------------------------
                         1/2                 1/2                  1/2
         ((j - m - 1)!)      ((j + m + 1)!)      ((2 j + 19)!)
```

Notice that one can modify this program to calculate, e.g., $w3f(j2, j, j1, m2, m, ml)$, $w3f(j1, j2, j, m1, m2, m)$, $w3f(j, j2, j1, m, m2, m1), \ldots$ by using the symmetry properties of the 3-j symbol.

*B. Input and Output for 6-*j *Symbols*

From Section 2.2, b is a nonnegative integer or half-integer, $d = f + \mu$ and $c = a + \nu$. Here, $-b \leq \mu,\ \nu \leq b$. Then, we compute

$$\begin{Bmatrix} a & b & c \\ \frac{1}{2} & c + \frac{1}{2} & b + \frac{1}{2} \end{Bmatrix},$$

$$\begin{Bmatrix} a & b & c \\ 1 & c + 1 & b - 1 \end{Bmatrix}$$

and

$$\begin{Bmatrix} 6 & 7 & 3 \\ \frac{13}{2} & \frac{13}{2} & \frac{9}{2} \end{Bmatrix}.$$

In the Maple V system (all inputs are case sensitive):

```
> read'6jfunm';
delta :=
  proc(a,b,c) local dl; dl := sqrt((a+b-c)!*(a-b+c)!/
    (a+b+c+1)!*(-a+b+c)!) end
w6f :=
proc(e,a,f,b,d,c)
local s0,s1,aa,v,vmin,vmax;
  aa := (a+b+c+1)!*(b+d+f+1)!/(a+b-c)!/(c-d+e)!/(c+d-e)!/
    (-e+a+f)!/(e-a+f)!/(b+d-f)!;
  v := min(2*b,-a+b+c,b-d+f);
  s0 := 0;
  if
  type(a,numeric) and type(b,numeric) and type(c,numeric) and
    type(d,numeric) and type(e,numeric)
    and type(f,numeric) then
      vmax := min(2*b,b+c-e+f,-a+b+c,b-d+f);
      for n from 0 to vmax do
        s0 := s0+(-1)^n*(2*b-n)!/n!*(b+c-e+f-n)!/
          (-a+b+c-n)!*(b+c+e+f+1-n)!/(b-d+f-n)!/
          (a+b+c+1-n)!/(b+d+f+1-n)!
      od;
      s1 := (-1)^(b+c+e+f)*simplify(
        radsimp(delta(a,b,c)*delta(c,d,e)*delta(a,e,f)
          *delta(b,d,f)*aa*s0))
    else
      for n from 0 to v do
        s0 := s0+(-1)^n*(2*b-n)!/n!*(b+c-e+f-n)!/
          (-a+b+c-n)!*(b+c+e+f+1-n)!/(b-d+f-n)!/
          (a+b+c+1-n)!/(b+d+f+1-n)!
      od;
      s1 := (-1)^(b+c+e+f)*delta(a,b,c)*delta(c,d,e)*
        delta(a,e,f)*delta(b,d,f)*aa*s0;
      s1 := factor(expand(s1))
  fi
end
```

```
> w6f(a,b,c,1/2,c+1/2,b+1/2);
```

$$-\frac{(b+1+c-a)^{1/2}\,(2+b+a+c)^{1/2}\,(-1)^{b}\,(-1)^{a}\,(-1)^{c}}{(2\,b+1)^{1/2}\,(2\,b+2)^{1/2}\,(2\,c+1)^{1/2}\,(2+2\,c)^{1/2}}$$

```
> w6f(a,b,c,1,c+1,b-1);
```

$$1/2\,(-1)^{b}\,(-1)^{a}\,(-1)^{c}\,2^{1/2}\,(b-c+a-1)^{1/2}\,(b-c+a)^{1/2}\,(-b+c+a+1)^{1/2}\,(-b+2+c+a)^{1/2}$$
$$\Big/\,\left((2\,b-1)^{1/2}\,b^{1/2}\,(2\,b+1)^{1/2}\,(2\,c+1)^{1/2}\,(2+2\,c)^{1/2}\,(3+2\,c)^{1/2}\right)$$

```
> w6f(6,7,3,13/2,13/2,9/2);
```

$$\frac{333}{68068}\,19^{1/2}\,3^{1/2}$$

```
> ifactor(op(1,"));
```

$$\frac{(3)^{2}\,(37)}{(2)^{2}\,(7)\,(11)\,(13)\,(17)}$$

The results agree with those reported in [1, 3–5]. In the computation of 6-j symbols, for large value of b in Eq. (7), the 6-j symbol outputs exhibit many terms. This can be avoided, while saving in CPU times, by deleting the *expand* command and keeping the factorials unexpanded. (This provides a more convenient formula for tabulation.)

*C. Input and Output for 9-*j *Symbols*

As we already showed in Section 2.3, j_{13}, j_{23}, and j_{33} must be integers or half-integers and $j_{11} = j_{12} + \lambda$, $j_{21} = j_{22} + \mu$ and $j_{31} = j_{32} + \nu$. For example, we then compute

$$\begin{pmatrix} a+\frac{1}{2} & b+\frac{1}{2} & c+1 \\ a & b & c \\ \frac{1}{2} & \frac{1}{2} & 1 \end{pmatrix},$$

$$\begin{pmatrix} a+2 & b+1 & c+1 \\ a & b & c \\ 2 & 1 & 1 \end{pmatrix},$$

and

$$\begin{pmatrix} 2 & 4 & 6 \\ 4 & 6 & 8 \\ 6 & 8 & 10 \end{pmatrix}.$$

In the Maple V System (all inputs are case-sensitive):

```
> read'9jfunm';
laplac := proc(a,b,c)
          local lap;
            lap := sqrt((a-b+c)!*(a+b-c)!*(a+b+c+1)!/
             (-a+b+c)!)
          end

w9f :=

   proc(j11,j21,j31,j12,j22,j32,j13,j23,j33)
   local x,y,z,x1,y1,z1,s0,s1,s2;
     s0 := (-1)^(j13+j23-j33)*laplac(j21,j11,j31)*laplac
       (j12,j22,j32)/laplac(j21,j22,j23)/laplac(j12,j11,j13)
       *laplac(j33,j31,j32)/laplac(j33,j13,j23);
     u1 := min(2*j23,j22-j21+j23,j13+j23-j33);
     s1 := 0;
     if type(j11,numeric) and type(j12,numeric) and
       type(j13,numeric) and type(j21,numeric) and
       type(j22,numeric) and type(j23,numeric) and
       type(j31,numeric) and type(j32,numeric) and
       type(j33,numeric) then
       x1 := min(j22-j21+j23,j13+j23-j33);
       y1 := min(j31-j32+j33,j12+j22-j32);
       z1 := min(j11-j12+j13,j11+j21-j31);
       for x from 0 to x1 do
         for y from 0 to y1 do
           for z from 0 to z1 do
             if 0 <= -j11+j12+j33-j23+x+z
               and 0 <= j21-j12+j32-j23+x+y
               and 0 <= j11+j21-j32+j33-y-z then
               s1 := s1+(-1)^(x+y+z)*(2*j23-x)!*
                 (j21+j22-j23+x)!/x!/
                 (j22-j21+j23-x)!/(j13+j23-j33-x)!/
                 (j21-j12+j32-j23+x+y)!*(j13-j23+j33+x)!*
                 (j22-j12+j32+y)!/(-j11+j12+j33-j23+x+z)!
                 /y!/(j12+j22-j32-y)!*(j31+j32-j33+y)!*
                 (j11+j21-j32+j33-y-z)!/(j31-j32+j33-y)!/
                 (2*j32+1+y)!/z!/(j11+j21-j31-z)!*(2*j11-z)!
```

```
                 *(j12-j11+j13+z)!/(j11-j12+j13-z)!/
                 (j11+j21+j31+1-z)!
             fi
           od
         od
       od;
       s2 := simplify(radsimp(s0)*s1)
     else
       for x from 0 to u1 do
         for y from 0 to j31-j32+j33 do
           for z from max(j11-j12-j13,0) to j11-j12+j13 do
             if 0 <= -j11+j12+j33-j23+x+z then
               s1 := s1+(-1)^(x+y+z)*(2*j23-x)!*
                 (j21+j22-j23+x)!/x!/
                 (j22-j21+j23-x)!/(j13+j23-j33-x)!/
                 (j21-j12+j32-j23+x+y)!*(j13-j23+j33+x)!*
                 (j22-j12+j32+y)!/(-j11+j12+j33-j23+x+z)!/
                 y!/(j12+j22-j32-y)!*(j31+j32-j33+y)!*
                 (j11+j21-j32+j33-y-z)!/(j31-j32+j33-y)!/
                 (2*j32+1+y)!/z!/(j11+j21-j31-z)!*(2*j11-z)!
                 *(j12-j11+j13+z)!/(j11-j12+j13-z)!/
                 (j11+j21+j31+1-z)!
             fi
           od
         od
       od;
       s2 := factor(expand(s0*s1))
     fi
   end

 > w9f(a+1/2,b+1/2,c+1,a,b,c,1/2,1/2,1);

                        1/2    1/2    1/2                        1/2                   1/2
1/6 (b + 1 - a + c)       2      6       (- b + 1 + a + c)          (b + 3 + a + c)

                      1/2   /              1/2             1/2              1/2             1/2
 (a + 2 + b + c)           /  ((2 b + 1)      (2 b + 2)       (2 a + 2)        (2 c + 1)
                          /

                                       1/2             1/2              1/2
                          (2 a + 1)       (2 c + 2)       (3 + 2 c))

 > w9f(a+2,b+1,c+1,a,b,c,2,1,1);

                                   1/2                      1/2                     1/2
        1/30 (a + 1 + b - c)          (a + 2 + b - c)          (b + a + c + 4)
```

```
                 1/2    1/2    1/2                     1/2                   1/2
(b + 5 + a + c)      6     30     (-b + 2 + a + c)       (-b + 1 + a + c)

                 1/2                 1/2  /              1/2          1/2
(b + 3 + a + c)     (a + 2 + b + c)      /  ((2 b + 1)      (2 b + 2)
                                        /

         1/2          1/2          1/2          1/2          1/2
(2 b + 3)    (2 a + 2)    (2 a + 3)    (2 a + 4)    (2 a + 5)

       1/2          1/2          1/2          1/2
(3 + 2 c)    (2 a + 1)    (2 c + 2)    (2 c + 1))
```

```
> w9f(2,4,6,4,6,8,6,8,10);
                         37903
                      - --------
                        97274034

> ifactor(");
                        (29) (1307)
                 - -------------------------------
                          4          2
                    (2)(3)   (11)(13)   (17) (19)
```

Again, the results agree with those reported in [1,3].

D. Outputs in Fortran and LaTex

The above outputs of 3-*j*, 6-*j*, and 9-*j* symbols can be translated into a Fortran format for other computational purposes or they can be converted into a LaTex format. Here is an example, in the Maple system:

```
> read'6jfunm';
> fortran(w6f(a,b,c,1,c-1,b-1));
    t0 = (-1)**b*(-1)**a*(-1)**c/sqrt(2*b-1)/sqrt(b)/sqrt
    #(2*b+1)*sqrt(b+c-a-1)*sqrt(-a+b+c)*sqrt(b+a+c)*sqrt
    #(1+b+a+c)/sqrt(1+2*c)/sqrt(2*c-1)/sqrt(c)/2
> latex(w6f(a,b,c,1,c-1,b-1));
{\frac {\left (-1\right )^{b}\left (-1\right )^{a}\left
(-1\right )^{c}\sqrt {b+c-a-1}\sqrt {-a+b+c}\sqrt {b+a+c}\sqrt
{1+b+a+c}}{2\,\sqrt {2\,b-1}\sqrt {b}\sqrt {2\,b+1}\sqrt {1+2\,c}
\sqrt{2\,c-1}\sqrt {c}}}
```

This output can be rewritten as

$$\begin{Bmatrix} a & b & c \\ 1 & c-1 & b-1 \end{Bmatrix} =$$

$$\frac{(-1)^b(-1)^a(-1)^c\sqrt{b+c-a-1}\sqrt{-a+b+c}\sqrt{b+a+c}\sqrt{1+b+a+c}}{2\sqrt{2b-1}\sqrt{b}\sqrt{2b+1}\sqrt{1+2c}\sqrt{2c-1}\sqrt{c}}.$$

4. Conclusions

We have pointed out algebraic formulas and numerically exact computations of 3-j, 6-j, and 9-j symbols by $3jfunm$, $6jfunm$, and $9jfunm$ Maple programs. These three programs reproduce the algebraic tables of 3-j, 6-j, and 9-j symbols [2–5]. If the values of the angular momenta (such as j_2 in 3-j, b in 6-j, and α, β, and γ in 9-j) are large, the *expand* command should be deleted from the programs. Otherwise, the computation takes a longer CPU time or may even fail, depending of the RAM memory and the variable disk space. For low values of the angular momentum, with a $16MB$ RAM memory, in a PC-386DX-40 MHz, the computations take only a few seconds. Additional examples are given in the Appendix. The produced algebraic tables are frequently needed in applications of angular momentum theory to physical problems.

Acknowledgments

The author is grateful to Professor Y. N. Chiu, Professor P. B. Macedo, and Dr. I. L. Pegg for many helpful discussions, constant support, and encouragement.

Appendix A: Test Run Outputs of 3-j Symbols

```
> read'3jfunm':
> for i from 0 to 2 do
> print('w3f(',j+2,j,2,m,-m-i,i,')=',
  w3f(j+2,j,2,m,-m-i,i));
> od;
  w3f(, j + 2, j, 2, m, - m, 0, )=,
```

$$\frac{(j+m+1)^{1/2}\,(j+m+2)^{1/2}\,(j+1-m)^{1/2}\,(j+2-m)^{1/2}\,6^{1/2}\,(-1)^{j}}{(2j+1)^{1/2}\,(2j+2)^{1/2}\,(2j+3)^{1/2}\,(2j+4)^{1/2}\,(2j+5)^{1/2}\,(-1)^{m}}$$

```
  w3f(, j + 2, j, 2, m, - m - 1, 1, )=,
```

$$2\,\frac{(j-m)^{1/2}\,(j+m+2)^{1/2}\,(j+1-m)^{1/2}\,(j+2-m)^{1/2}\,(-1)^{j}}{(2j+1)^{1/2}\,(2j+2)^{1/2}\,(2j+3)^{1/2}\,(2j+4)^{1/2}\,(2j+5)^{1/2}\,(-1)^{m}}$$

```
  w3f(, j + 2, j, 2, m, - m - 2, 2, )=,
```

$$\frac{(j-m-1)^{1/2}\,(j-m)^{1/2}\,(j+1-m)^{1/2}\,(j+2-m)^{1/2}\,(-1)^{j}}{(2j+1)^{1/2}\,(2j+2)^{1/2}\,(2j+3)^{1/2}\,(2j+4)^{1/2}\,(2j+5)^{1/2}\,(-1)^{m}}$$

```
> for i from 0 to 2 do
> print('w3f(',j+1,j,2,m,-m-i,i,')=',
```

```
  w3f(j+1,j,2,m,-m-i,i));
> od;

  w3f(, j + 1, j, 2, m,- m, 0, )=,

                   1/2              1/2    1/2  1/2        1/2
          (j + m + 1)    (j + m - 1)      6      2      (-1)       m
      - ---------------------------------------------------------------
         1/2           1/2           1/2           1/2           1/2       m
        j     (2 j + 1)     (2 j + 2)     (2 j + 3)     (2 j + 4)     (-1)

  w3f(, j + 1, j, 2, m, - m - 1, 1, )=,

                                         1/2           1/2   1/2      j
          (j + 2 m + 2) (j + 1 - m)     (j - m)       2     (-1)
      - ---------------------------------------------------------------
             m             1/2           1/2           1/2           1/2  1/2
         (-1)     (2 j + 4)     (2 j + 3)     (2 j + 2)     (2 j + 1)    j

  w3f(, j + 1, j, 2, m, - m - 2, 2, )=,

                     1/2         1/2               1/2               1/2   1/2       j
       (j - m - 1)     (j - m)     (j + m + 2)     (j + 1 - m)     2     (-1)
     - -----------------------------------------------------------------------
          1/2           1/2           1/2           1/2           1/2       m
         j     (2 j + 1)     (2 j + 2)     (2 j + 3)     (2 j + 4)     (-1)

> for i from 0 to 2 do
> print('w3f(',j,j,2,m,-m-i,i,')=',w3f(j,j,2,m,-m-i,i));
> od;

   w3f(, j, j, 2, m, -m, 0, )=,

                            1/2       j   2         2
                           2     (-1)   (j + j - 3 m)
     - -------------------------------------------------------------------
                      1/2   1/2           1/2           1/2           1/2      m
          (2 j - 1)      j     (2 j + 1)     (2 j + 2)     (2 j + 3)     (-1)

  w3f(, j, j, 2, m, - m - 1, 1, )=,

                1/2          1/2   1/2       j                              1/2
               6     (j - m)      2     (-1)   (2 m + 1) (j + m + 1)
      1/2 -----------------------------------------------------------------
                           1/2   1/2           1/2           1/2           1/2      m
              (2 j - 1)       j     (2 j + 1)     (2 j + 2)     (2 j + 3)     (-1)

  w3f(, j, j, 2, m, - m - 2, 2, )=,

            1/2               1/2          1/2               1/2               1/2   1/2       j
           6     (j - m - 1)     (j - m)     (j + m + 1)     (j + m + 2)      2     (-1)
  1/2 ---------------------------------------------------------------------------------
                              1/2   1/2           1/2           1/2           1/2      m
              (2 j - 1)          j     (2 j + 1)     (2 j + 2)     (2 j + 3)     (-1)
```

Appendix B: Test Run Outputs of 6-*j* Symbols

```
> read'6jfunm':
> for p from 1/2 to 3/2 do
> print('w6f(',a,b,c,3/2,c-p,b-p,')=',w6f(a,b,c,3/2,
  c-p,b-p));
> od;
```

w6f(, a, b, c, 3/2, c − 1/2, b − 1/2,)=,

$$1/2\,(-a+c+b)^{1/2}\,(-1)^{b}\,(-1)^{c}\,(-1)^{a}\,(b+c+a+1)^{1/2}\,(3a^{2}+3a-3c^{2}+2-b^{2}-3b+2bc-c)\Big/\big((2b-1)^{1/2}\,b^{1/2}\,(2b+1)^{1/2}\,(2b+2)^{1/2}\,(2c-1)^{1/2}\,c^{1/2}\,(1+2c)^{1/2}\,(2+2c)^{1/2}\big)$$

w6f(, a, b, c, 3/2, c − 3/2, b − 3/2,)=,

$$1/2\,(-1)^{b}\,(-1)^{c}\,(-1)^{a}\,(b+c-a-2)^{1/2}\,(b-1+c-a)^{1/2}\,(-a+c+b)^{1/2}\,(b+a+c-1)^{1/2}\,(b+a+c)^{1/2}\,(b+c+a+1)^{1/2}\Big/\big((2b-2)^{1/2}\,(2b-1)^{1/2}\,b^{1/2}\,(2b+1)^{1/2}\,(-2+2c)^{1/2}\,(2c-1)^{1/2}\,c^{1/2}\,(1+2c)^{1/2}\big)$$

Appendix C: Test Run Outputs of 9-*j* Symbols

```
> read'9jfunm':
> w9f(a+3/2,b+1/2,c+1,a,b,c,3/2,1/2,1);
```

$$1/12\,(b+1-c+a)^{1/2}\,(b+4+c+a)^{1/2}\,6^{1/2}\,2^{1/2}\,3^{1/2}\,(-b+2+c+a)^{1/2}\,(-b+c+a+1)^{1/2}\,(a+3+b+c)^{1/2}\,(2+b+a+c)^{1/2}\Big/$$

$$((2b+1)^{1/2}\,(2b+2)^{1/2}\,(2a+3)^{1/2}\,(2a+4)^{1/2}\,(3+2c)^{1/2}$$

```
                 1/2           1/2           1/2           1/2
        (2 a + 2)     (2 a + 1)     (1 + 2 c)     (2 + 2 c)     )

> w9f(a+2,b+2,c+1,a,b,c,2,2,1);
                            1/2                 1/2                  1/2
       1/15 (b + 1 + c - a)    (b + 1 - c + a)    (a + 2 + b - c)

                      1/2                1/2                 1/2                1/2
    (b + 3 - c + a)    (b + 4 + c + a)    (b + c + a + 5)    (b + 6 + c + a)

     1/2  1/2                1/2                   1/2                 1/2 /
    2    30   (2 + b + a + c)   (-b + c + a + 1)   (a + 3 + b + c)     /
                                                                      /

                 1/2         1/2          1/2          1/2          1/2
       ((2 b + 1)   (2 b + 2)    (2 b + 3)    (2 b + 4)    (2 b + 5)

                 1/2         1/2          1/2          1/2          1/2
       (2 a + 2)    (2 a + 3)    (2 a + 4)    (2 a + 5)    (1 + 2 c)

                 1/2         1/2          1/2
       (2 a + 1)    (2 + 2 c)    (3 + 2 c)    )
```

Bibliography

[1] S. T. Lai, I. L. Pegg, Y. N. Chiu, and V. M. Bogdan, submitted.

[2] R. N. Zare, *Angular Momentum* (Wiley, New York, 1988).

[3] D. A. Varshalovich, A. N. Moskalev, and V. K. Khersonskii, *Quantum Theory of Angular Momentum* (World Scientific, Singapore, 1989).

[4] A. R. Edmonds, *Angular Momentum in Quantum Mechanics,* 2nd ed. (Princeton University Press, Princeton, NJ, 1960).

[5] L. C. Biedenharn and J. D. Louck, *Angular Momentum in Quantum Physics, Theory and Applications, Encyclopedia of Mathematics and Its Applications,* Vol. 8 (Addison-Wesley, Reading, MA, 1981).

[6] E. P. Wigner, *Group Theory and Its Application to the Quantum Mechanics of Atomic Spectra* (Academic Press, New York, London, 1959).

[7] G. Racah, Phys. Rev. **62**, 438 (1942).

[8] B. L. Van der Waerden, *Die Gruppentheoretische Methode in der Quantenmechanik* (Springer, Berlin, 1932).

[9] S. D. Majumdar, Prog. Theor. Phys. **20**, 798 (1958).

[10] A. P. Jucys and A. A. Bandzaitis, *Theory of Angular Momentum in Quantum Mechanics,* 2nd ed. (Mintus, Vilnius, 1977).

Received March 31, 1994
Accepted for publication April 20, 1994